

地球惑星科学実習 B: 常微分方程式の数値解法

岩山 隆寛 *

2016 年 11 月 4, 11, 18, 25 日

1 はじめに

物理現象の発展を記述する方程式は, Newton の運動方程式, Maxwell 方程式, Navier-Stokes 方程式など, 常微分方程式系や偏微分方程式系の形で書かれることが多いことはよく知られている. この回では, Newton の運動方程式のような常微分方程式系を計算機を用いて解くことを扱う. 解析的に解くことができない (もしくはできそうにない) 常微分方程式や偏微分方程式に遭遇した時にこの回やその次に紹介する微分方程式の数値解法が強力な手段となる.

Newton の運動方程式は質点の位置を未知変数とすると 2 階の微分方程式である. しかし, 運動量を導入すると, 1 階の連立微分方程式に書き下すことができる. そこで先ず 1 階の単一の常微分方程式,

$$\frac{dx}{dt} = f(x, t), \quad (1)$$

をある初期条件^{*1}(時間 $t = t_0 (= 0)$ における $x(t)$ の値),

$$x(t_0) = x_0, \quad (2)$$

のもとで数値的に解くことを考える. f は既知の関数である.

数値計算では連続的な変数は扱えないので, 時間 t は t_0, t_1, t_2, \dots というとびとびの値になり,

$$\Delta t_1 = t_1 - t_0, \Delta t_2 = t_2 - t_1, \dots, \Delta t_i = t_i - t_{i-1}, \dots \quad (3)$$

* 神戸大学 大学院理学研究科 惑星学専攻. e-mail: iwayama@kobe-u.ac.jp

^{*1} (1) における t は時間でなくてもよいがここでは時間と呼ぶことにする.

は刻み幅、もしくはステップ幅という。以下では、 Δt_i は i によらず一定値 Δt であるとする。

初期条件が与えられた時に、時間 $t = t_1, t_2, t_3, \dots$ における $x(t)$ の値、 $x(t_1), x(t_2), x(t_3), \dots$ を求めていく。計算機によって求められたこれら $x(t_i)$ の値を x_i と表すことにする。(1) は形式的に積分すると、

$$x(t_i) = \int_{t_0}^{t_i} f(x(t), t) dt, \quad (4)$$

とかける。即ち、 $x(t)$ を求めるには積分する必要がある。前回までに学んだ数値積分との違いは、(1) の右辺の f に x が含まれるか、含まれないかの違いである。もし、(1) の右辺の f に x が含まれないのであれば、(1) の問題は単に定積分、

$$x(t_i) = \int_{t_0}^{t_i} f(t) dt, \quad (5)$$

であり、前回までに学んだ知識で解ける。ここでは、 f に未知変数 x を含む場合を考える。

数値積分の際に、素朴な計算法から議論を出発して、より精度の高い計算法を学んだ。ここでも常微分方程式の素朴な数値計算法から出発して、より精度の高い計算法を導入していくことにする。

2 Euler 法を用いた 1 階常微分方程式の解法

2.1 解説

独立変数 t を $\Delta t > 0$ の一定の刻み幅で離散化し、計算機で表現する時刻を

$$t_k = k\Delta t, \quad k = 1, 2, 3, \dots \quad (6)$$

と表す。計算機によって求められた $x(t_k)$ の値を x_k と表すことにする。微分の定義を参照すると、

$$\left(\frac{dx}{dt}\right)_{t=t_k} = \lim_{\Delta t \rightarrow 0} \frac{x(t_{k+1}) - x(t_k)}{\Delta t}, \quad (7)$$

である。計算機では $\Delta t \rightarrow 0$ の極限は取れないので、(7) の右辺の極限を無視し、

$$\left(\frac{dx}{dt}\right)_{t=t_k} \approx \frac{x_{k+1} - x_k}{\Delta t}, \quad (8)$$

と近似し, (1) を

$$x_{k+1} = x_k + f(x_k, t_k)\Delta t \quad (9)$$

と変形する. (9) に従って常微分方程式を解く方法を Euler 法という. (8) のような微分の差分による置き換えは, 前進差分と呼ばれる.*2 (9) は常微分方程式の数値解法の中で最も簡単なものである. 初期条件 t_0, x_0 が与えられるとそれらを用いて $f(x_0, t_0)$ を求め, (9) にしたがって, t_1 における x_1 が求まる. さらに, t_1, x_1 から (9) に従って t_2 における x_2 が計算でき, ... と次々に将来の時間における x の値が計算できる.

2.2 課題

次の常微分方程式 ($\lambda > 0$ とする) の解析解を求めなさい. さらに Euler 法により数値解を求めて, 両者を比較しなさい.

$$\frac{dx}{dt} = -\lambda x, \quad x(0) = x_0. \quad (12)$$

数値解では, λ, x_0 の値を自分で適当な数値において計算しなさい. 課題の提出方法は, サンプルを参照しなさい.

2.3 サンプルプログラム

(12) を Euler 法を用いて数値的に解くプログラムの例を以下に示す. 以下では, $\lambda = 1$, $\Delta t = 10^{-3}$, 初期条件 $x_0 = 1$ のもとで (12) を $t_{\max} = 10$ まで解いている. 全ての時間で x の値を出力するのではなく, $t_{\text{out}} = 10^{-1}$ 毎に x の値を x_{out} という配列に格納していき, 最後に装置番号 10 で指定した results.dat というファイルに t と x の値を書き出している. 書き出しの形式は format 文で指定されており, 符号, 仮数部, 指数部を合わせて 12 ケタで, 仮数部を少数以下 5 桁の実数となるように指定している.

*2 他にも後退差分,

$$\left(\frac{dx}{dt}\right)_{t=t_k} \approx \frac{x_k - x_{k-1}}{\Delta t}, \quad (10)$$

や中央差分,

$$\left(\frac{dx}{dt}\right)_{t=t_k} \approx \frac{x_{k+1} - x_{k-1}}{2\Delta t}, \quad (11)$$

と呼ばれる微分の近似表現がある.

```

c sample program for numerical calculation of an ordinary differential equation
c by the Euler method
c produced by Takahiro IWAYAMA
c 2015.05.27, 2016.11.03
c23456789012345678901234567890123456789012345678901234567890123456789012
  program ode_1

  implicit none
  integer i, k, k_max, k_out
  real*8 t, t_max, dt, t_0, t_out
  real*8 x, dx, x_0
  parameter (t_max=10.0d0, dt=1.0d-3, t_out=1.0d-1)
  real*8 x_out(0:int(t_max/t_out)), f

  k_max=int(t_max/dt)
  k_out=int(t_out/dt)
  k=0

c initial condition
  t_0=0.d0
  x_0=1.d0

  t=t_0
  x=x_0
  x_out(k)=x

c time evolution
  do 10 i=1,k_max
    t=t+dt
c differential coefficient * dt
    dx=f(x,t)*dt
c evloutuion

```

```

        x=x+dx
c output control
        if (mod(i,k_out)==0) then
            k=k+1
            x_out(k)=x
        endif
10    continue

        open(10,file='ode_1.dat')
        do 20 k=0,int(t_max/t_out)
            write(10,'(1x,E12.5,1x,E12.5)') t_0+k*k_out*dt, x_out(k)
20    continue
        close(10)

        stop
        end

c-----
        real*8 function f(x,t)

        real*8  x,t,lambda
        lambda=1.0D+0

        f=-lambda*x

        return
        end

```

2.4 結果

(12) を Euler 法により数値的に解いた結果と解析解 ($x(t) = Ce^{-t}$, ここで C は任意定数) とを図 1 に示す. 数値計算の条件は, 時間刻み $\Delta t = 10^{-3}$, 初期条件 $x(0) = 1$, $\lambda = 1$ で $t = 10$ まで倍精度で解いている. (初期条件から $C = 1$ である.) 図 1 からわかるよう

に、数値解は解析解をよく表現できていることがわかる。^{*3}




図1 Euler法による数値計算で求めた(12)の解(+印)とその解析解(破線)との比較.

2.5 Gnuplot の使い方についてのメモ

- x 軸のラベルを t にする.

```
gnuplot> set xlabel "t"
```

- y 軸のラベルを $x(t)$ にする.

```
gnuplot> set ylabel "x(t)"
```

- results.dat にあるデータと $\exp(-t)$ の関数形を表示し、右上の凡例を results.dat は Euler と表示し、 $\exp(-t)$ は theory と表示する.

^{*3} 図示して目で比べるのではなく、解析解と数値解の相対誤差を時間の関数として表現してもよい。余力のある人は挑戦してみてください。

```
gnuplot> p 'results.dat' title 'Euler', exp(-x) title 'exp(-t)'
```

- 図の出力を画面ではなく tex ファイルに張り込むために eps という形式のファイルに出力する。その際、軸のラベルなどの文字は 20 ポイントの大きさにする。さらに図をカラーで保存する。

```
gnuplot> set terminal postscript eps 20 enhanced color
```

- 図の出力先のファイルとして figure.eps を指定する。

```
gnuplot> set output "figure.eps"
```

- 図を再描画する

```
gnuplot> replot
```

2.6 T_EX の使い方についてのメモ

前小節の図 1 のように T_EX の文章に図を張る場合には、図を eps 形式の図（たとえば ode1.eps）を用意する。T_EX の文章のプリアンブルを次のようにする：

```
\documentclass[a4]{jsarticle}
\usepackages[dvipdfmx]{graphicx} <-これを追加する.
```

さらに T_EX 内の図を張り込みたい場所に、

```
\begin{figure}[htbp]
  \begin{center}
    \includegraphics[width=7cm,angle=270]{ode1.eps}
    \vspace{2zw}
    \caption{図の説明.}\label{fig_1}
  \end{center}
\end{figure}
```

と記述する.

2.7 宿題

あなたの好きな 1 階の常微分方程式を挙げなさい. その解析解^{*4}を示し, さらにその微分方程式を Euler 法で数値的に解き, 解析解と数値解を比較しなさい.^{*5}

3 Euler 法を用いた 2 階常微分方程式の解法

3.1 解説

ここでは空間 1 次元問題を考える. 質量 m の質点が力 F の作用の下で運動する場合の Newton の運動方程式,

$$m \frac{d^2x}{dt^2} = F(x, \dot{x}, t), \quad (13)$$

は, 時間に関する 2 階の微分方程式である. ここで, 質点に働く F は質点の位置と速度 $\dot{x} \equiv \frac{dx}{dt}$, 時間 t に依存するとした. 運動量 $p \equiv m\dot{x}$ を導入すれば, (13) は

$$\frac{dx}{dt} = \frac{p}{m}, \quad (14)$$

$$\frac{dp}{dt} = F\left(x, \frac{p}{m}, t\right), \quad (15)$$

と 1 階の連立常微分方程式として書ける. そこで, 2 階の常微分方程式も 2 節で述べた 1 階の常微分方程式の数値解法の知識を使えば, 数値的に解けることになる.

3.2 演習問題

次の常微分方程式の解析解を求めなさい. さらに Euler 法により数値解を求めて, 両者を比較しなさい.

ばね定数 k の線形ばねに繋がれた質量 m の質点の運動方程式は次の 2 階の微分方程式,

$$m \frac{d^2x}{dt^2} = -kx, \quad (16)$$

^{*4} 注意: 微分方程式を解析的に解いた, 手で解いた, 解を解析解と呼びます. 初期条件を事前に指定していなければ解析解は微分方程式の一般解で任意定数を含みます. 任意定数は初期条件によって決定されます. 微分方程式を計算機を使って解くときには, 初期条件を指定しないと解けませんので得られる解は特殊解 (任意定数に具体的な値が代入された解) です.

^{*5} 地球惑星科学基礎 I や III で 1 階の常微分方程式を解きました. 参考にするといいでしょう.

に書ける. これを $m = 1, k = 1$, 初期条件, $x(0) = 1, dx(0)/dt = 0$, のもとで解くことを考える. このとき, (16) は以下のような 1 階の連立常微分方程式に書き直すことができる:

$$\frac{dx}{dt} = p, \quad x(0) = 1. \quad (17)$$

$$\frac{dp}{dt} = -x, \quad p(0) = 0. \quad (18)$$

ここで, p は運動量である.

3.3 ヒント

1 階の単一の常微分方程式を解くプログラムを利用して,

$$\frac{dx}{dt} = G(x, p, t), \quad (19)$$

$$\frac{dp}{dt} = F(x, p, t), \quad (20)$$

を解くプログラムを書く. そのための作業の概要は, 次のとおりです:

1. 新しい変数として p とそれに関連する変数を導入する.
2. 新しい変数 p の初期条件を設定する.
3. ファイルに書きだす変数は, t, x, p とする.
4. 上記の変更に合わせて, FORMAT 文を変更する.
5. 関数 F を x, t を変数とする関数から, x, p, t を変数とする関数に書き換える.
6. 新たに $G(x, p, t)$ という関数を導入する.

3.4 サンプルプログラム

以下にサンプルプログラムを添付する.

```
c sample program for numerical integration of an ordinary differential equations
c by the Euler method
c produced by Takahiro IWAYAMA
c 2015.05.27-0605, 20161105
c23456789012345678901234567890123456789012345678901234567890123456789012
  program ode_2

  implicit none
  integer i, k, k_max, k_out
  real*8 t, t_0, dt, t_out, t_max
  real*8 x, x_0, dx
  real*8 p, p_0, dp
  parameter (t_max=40.0d0, dt=1.0d-3, t_out=1.0d-1)
  real*8 x_out(0:int(t_max/t_out)), f
  real*8 p_out(0:int(t_max/t_out)), g

  k_max=int(t_max/dt)
  k_out=int(t_out/dt)
  k=0

c initial condition
  t_0=0.d0
  x_0=1.d0
  p_0=0.d0

  t=t_0
  x=x_0
  p=p_0
```

```

        x_out(k)=x
        p_out(k)=p

c time evolution
        do 10 i=1,k_max
            t=t+dt
c differential coefficients
            dx=g(x,p,t)*dt
            dp=f(x,p,t)*dt
c evloutuion
            x=x+dx
            p=p+dp
c output control
            if (mod(i,k_out)==0) then
                k=k+1
                x_out(k)=x
                p_out(k)=p
            endif
10    continue

        open(10,file='ode_2.dat')
        do 20 k=0,int(t_max/t_out)
            write(10,'(1x,E12.5,1x,E12.5,1x,E12.5)')
&      t_0+k*k_out*dt, x_out(k), p_out(k)
20    continue
        close(10)

        stop
        end

c-----
        real*8 function g(x,p,t)

```

```
real*8 x, p, t, m
m=dbl(1)
```

```
g=p/m
```

```
return
end
```

c-----

```
real*8 function f(x,p,t)
```

```
real*8 x, p, t, k
k=dbl(1)
```

```
f=-k*x
```

```
return
end
```

3.5 Gnuplot の使い方についてのメモ

- results.dat に 3 列のデータが格納されており, 各列は, t, x, p の値であるとする. t, p の時系列を gnuplot で表示したいときには, using とオプションを使う.

```
gnuplot> p 'results.dat' using 1:3
```

using を使用しないと単に 1 列目と 2 列目のデータが表示される.

3.6 結果

図 2, 3, 4 に (16) を上で指定された初期条件のもとで解いた解析解 ($x(t) = \cos(t), p(t) = -\sin(t)$) と Euler 法による数値解を示す. 図 2, 3 はそれぞれ x, p の時系列であるが, 図 4 は横軸, 縦軸が $x(t) - p(t)$ になっており, 位相空間と呼ばれる. 位

相空間は解析力学で登場する概念であるが, 長時間の解の軌道を見るのに便利な表記でもある. 図 2, 3 によると, 数値解と解析解は一致しているように見えるが, 図 4 を見ると振動の振幅が時間とともに増大していつていることがわかる.




図 2 Euler 法による数値計算で求めた (16) の解 (+ 印) とその解析解 (破線) の比較 ($x(t)$ の時系列).




図 3 Euler 法による数値計算で求めた (16) の解 (+ 印) とその解析解 (破線) の比較 ($p(t)$ の時系列).




図4 Euler法による数値計算で求めた(16)の解の位相空間の軌道. $t = 100$ までを
 図示している. 解析解は半径1の円であるが時間とともにわずかに半径が増大していっ
 ている.

3.7 宿題

あなたの好きな2階の常微分方程式を挙げ, その解析解を示しなさい. またその微分方
 程式を適当な初期条件のもとでEuler法によって数値的に解き, 解析解と数値解を比較し
 なさい.

4 Euler法の精度

常微分方程式を解くときには, 微分を差分で表現しているのでそれに伴う誤差,

$$e_k \equiv x(t_k) - x_k, \quad (21)$$

が存在する. Euler法の誤差を見積もってみる. ここで, $x(t_k)$ を解析解, x_k を数値解とす
 る. t_{k+1} におけるEuler法の誤差は,

$$e_{k+1} = x(t_{k+1}) - x_{k+1} = x(t_{k+1}) - \{x_k + f(x_k, t_k)\Delta t\}$$

である. ここで, $t = t_k$ においては解析解と数値解は一致している, $x_k = x(t_k)$, と仮定し, さらに $x(t_{k+1})$ を t_k の周りで Taylor 展開する:

$$\begin{aligned} e_{k+1} &= \left\{ x(t_k) + \Delta t \frac{dx(t_k)}{dt} + \frac{\Delta t^2}{2!} \frac{d^2x(t_k)}{dt^2} + \mathcal{O}(\Delta t^3) \right\} - \{x_k + f(x_k, t_k)\Delta t\} \\ &= \frac{\Delta t^2}{2!} \frac{d^2x(t_k)}{dt^2} + \mathcal{O}(\Delta t^3). \end{aligned} \quad (22)$$

つまり, Euler 法の誤差は $\mathcal{O}(\Delta t^2)$ であり, Euler 法は Δt の 1 次まで正しい計算法であることがわかる. そこで 1 次の公式とも呼ばれる. Euler 法における誤差は無限に続く Taylor 展開を Δt の 1 次の項で打ち切ったために生じたものであり, 打ち切り誤差であるといえる.

5 Adams-Bashforth 法

5.1 解説

2 節では 1 次の公式の Euler 法についてみた. より高次の公式を導こう. ここでは 2 次の公式である Adams-Bashforth 法を導く.

$x(t_{k+1})$ の Taylor 展開において, Δt の 2 次までの項を残すことを考える:

$$\begin{aligned} x(t_{k+1}) &= x(t_k) + \frac{dx(t_k)}{dt} \Delta t + \frac{1}{2!} \frac{d^2x(t_k)}{dt^2} \Delta t^2 + \mathcal{O}(\Delta t^3) \\ &= x(t_k) + f(x(t_k), t_k) \Delta t + \frac{1}{2!} \frac{df(x(t_k), t_k)}{dt} \Delta t^2 + \mathcal{O}(\Delta t^3) \\ &\approx x_k + f(x_k, t_k) \Delta t + \frac{1}{2} \frac{f(x_k, t_k) - f(x_{k-1}, t_{k-1})}{\Delta t} \Delta t^2 \\ &= x_k + \left(\frac{3}{2} f(x_k, t_k) - \frac{1}{2} f(x_{k-1}, t_{k-1}) \right) \Delta t. \end{aligned} \quad (23)$$

ここで f の微分は, 後方差分により近似した. 即ち,

$$x_{k+1} = x_k + \left(\frac{3}{2} f(x_k, t_k) - \frac{1}{2} f(x_{k-1}, t_{k-1}) \right) \Delta t \quad (24)$$

にしたがって常微分方程式を数値的に解く方法は Δt の 2 次まで正しく, この方法を Adams-Bashforth 法という. この方法では, 2 つの時間における関数の値 $f(x_k, t_k)$, $f(x_{k-1}, t_{k-1})$ が必要である. そこで t_0 から数値計算を始める場合, 何らかの方法 (例えば Euler 法) で t_1 における微分方程式の解を得ておく必要がある. その後の発展は, (24) に従って, t_2, t_3, \dots における x の値が計算できることになる.

5.2 演習問題

次の常微分方程式を Euler 法と, Adams-Bashforth 法により数値解を求めて, 解析解と比較しなさい. 特に, 図 4 のように位相空間で解の軌道を書いて確かめなさい.

ばね定数 k の線形ばねに繋がれた質量 m の質点の運動方程式は次の 2 階の微分方程式,

$$m \frac{d^2x}{dt^2} = -kx \quad (25)$$

に書ける. これを $m = 1, k = 1$, 初期条件, $x(0) = 1, dx(0)/dt = 0$, のもとで解くことを考える. (16) は以下のような 1 階の連立常微分方程式に書き直すことができる:

$$\frac{dx}{dt} = p, \quad x(0) = 1. \quad (26)$$

$$\frac{dp}{dt} = -x, \quad p(0) = 0. \quad (27)$$

このときの解析解は $x(t) = \cos t, p(t) = -\sin t$ である. したがって, 位相空間における解の軌道は $x^2 + p^2 = 1$ つまり, 原点を中心とした半径 1 の円となる.

5.3 ヒント

2 階の常微分方程式を Euler 法を用いて解くプログラムを利用する. そのための作業の概要は次のとおりです:

1. dt だけ以前の変数を記憶しておく必要がある. そこで, 新しい変数として `t_old`, `x_old`, `p_old` を宣言する.
2. `c_1`, `c_2` という倍精度の変数を宣言する. これらに `3/2`, `1/2` の値を代入しておく. (倍精度の数値の表現の仕方に注意.)
3. 最初のステップは Euler 法で計算する.
4. Euler 法で `x`, `p` が新しい時刻の値に書き換わる直前に, それらの値を `t_old`, `x_old`, `p_old` に代入しておく. 例えば変数 `x` については
`dx=g(x,p,t)*dt`
`t_old=t`
`x_old=x`
`x=x+dx` とする.
5. 第 2 ステップからは, (24) に従った計算式で新しい時刻の `x`, `p` を計算する.

6. このときにも x, p が新しい時刻の値に書き換わる直前に, それらの値を t_old, x_old, p_old に代入しておく. 例えば変数 x については,
- $$dx=(c_1*f(x,p,t)-c2*f(x_old,p_old,t_old))*dt$$
- $$t_old=t$$
- $$x_old=x$$
- $x=x+dx$ とする.

5.4 結果

図 5 に (25) を上で指定された条件のもとで Adams-Bashforth 法によって解いた数値解の位相空間の軌道を示す. Euler 法による解 (図 4) と異なり, 振動の振幅は 1 に保たれていることがわかる.




図 5 Adams-Bashforth 法による数値計算で求めた (25) の解の位相空間の軌道. $t = 100$ までを図示している.

5.5 Gnuplot の使い方についてのメモ

- gnuplot では図は縦横比が黄金比率になるように図が書かれる. これを変更して図 5 のように正方形にしたいときには以下のように指定してから図を描く.

```
gnuplot> set size square
```

とする.

5.6 宿題

あなたの好きな 2 階の常微分方程式を挙げ, その解析解を示しなさい. またその微分方程式を適当な初期条件のもとで Euler 法と Adams-Bashforth 法によって数値的に解き, 解析解と数値解を比較しなさい.

6 適合性, 安定性

微分方程式を差分で近似して解く場合に, 注意しておくべき事項, 2 点についてまとめた.

6.1 適合性

差分間隔 Δt を 0 にした極限で, 差分方程式は微分方程式に一致しなければならない. これを適合性という. 差分の切断誤差が Δt の 1 次以上であれば, その差分は適合性を満たす.

6.2 安定性

数値解の誤差 e_k が, Δt を固定して $k \rightarrow \infty$ ($t \rightarrow \infty$) のときに有界の場合, その差分による解法は安定である, という. もしくは, どのような初期条件に対しても安定な数値解をもつ場合, その差分による解法は安定である, という.

差分法が安定でない場合には, 時間積分の途中でオーバーフローを起こして望みの時間までの積分ができない. そのため, 差分法の安定性を調べることは実用上重要である.

以下では, 差分法の安定性を調べる方法の中で一番よく使われる方法, von Neumann の方法を紹介する. $t = t_k$ における微分方程式の解を x_k とする. $x_{k+1} = \lambda x_k$ として, 増幅因子 λ を導入する. 差分法が安定であるためには,

$$|x_k| = |\lambda| |x_{k-1}| = \dots = |\lambda|^k |x_0| < B, \quad (28)$$

ここで, B はある有限の数である. (28) を変形して,

$$k \ln |\lambda| < \ln \frac{B}{|x_0|} \equiv B'.$$

$k = t/\Delta t$ に注意すると,

$$\begin{aligned} \ln |\lambda| &< \frac{B'}{t} \Delta t \\ \therefore \ln |\lambda| &\leq \mathcal{O}(\Delta t). \end{aligned} \tag{29}$$

(29) より差分法が安定であるための条件として,

$$|\lambda| \leq 1 + \mathcal{O}(\Delta t) \tag{30}$$

を得る. (30) は von Neumann の条件と呼ばれる. 微分方程式の真の解が単調に増加するのでなければ, (30) は

$$|\lambda| \leq 1, \tag{31}$$

である.

6.3 例

1. 減衰の微分方程式

$$\frac{dx}{dt} = -ax, \quad (a > 0) \tag{32}$$

について, Euler 法と Adams-Bashforth 法の安定性を調べてみる.

Euler 法によると, (32) は

$$x_{k+1} = x_k - a\Delta t x_k \tag{33}$$

である. (33) に $x_k = \lambda x_{k-1}$ を代入して,

$$\lambda = 1 - a\Delta t. \tag{34}$$

$a\Delta t < 1$ ならば (34) は von Neumann の条件を満足している. 上記の結果を 2.2 節の計算と比較しなさい. サンプルプログラムでは, $a = 1$, $\Delta t = 10^{-3}$ の条件で解いているので, (34) より数値計算は安定に行えることが分かる. 一方 $a = 100$, $\Delta t = 10^{-1}$ では数値計算は安定に行えないことが予測される.

いっぽう, Adams-Bashforth 法によると, (32) は

$$x_{k+1} = x_k - \frac{a\Delta t}{2} (3x_k - x_{k-1}) \quad (35)$$

である. (35) に $x_k = \lambda x_{k-1}$ を代入して,

$$\lambda^2 = \lambda - \frac{a\Delta t}{2} (3\lambda - 1). \quad (36)$$

(34) は 2 つの根,

$$\lambda_+ = \frac{1}{2} \left\{ \left(1 - \frac{3a\Delta t}{2} \right) + \sqrt{1 - a\Delta t + \frac{9}{4}a^2\Delta t^2} \right\}, \quad (37a)$$

$$\lambda_- = \frac{1}{2} \left\{ \left(1 - \frac{3a\Delta t}{2} \right) - \sqrt{1 - a\Delta t + \frac{9}{4}a^2\Delta t^2} \right\}, \quad (37b)$$

$$(37c)$$

を持ち, $a\Delta t \ll 1$ で $O(a^2\Delta t^2)$ を無視すると, $\lambda_+ \approx 1 - a\Delta t$, $\lambda_- \approx -a\Delta t/2$ となる. λ_+ は Euler 法を適用した時に得られた増幅因子と同じものであるが, λ_- は差分化の手続きによって生じたモードの増幅因子である. $x_k^{(+)} = \lambda_+^k x_0^{(+)}$, $x_k^{(-)} = \lambda_-^k x_0^{(-)}$ は, それぞれ, 物理モード, 計算モードと呼ばれ, 差分方程式の解はこの線形結合で $x_k = \alpha x_k^{(+)} + \beta x_k^{(-)}$ と書ける. ここで, α, β は初期条件から決まる定数である. 計算モードが現れないようにするためには, 慎重に初期条件を選ばないといけないが, 解くべき方程式が複雑な場合には, ここで展開しているような解析は困難なので, 計算モードがゼロとなる初期条件を選ぶことはほとんど不可能である. また, 数値計算では丸め誤差を伴うので計算モードの除去は困難である. そこで, 計算モードの増幅率が負となる差分解法が望まれる.

減衰型の方程式に Adams-Bashforth 法を適用して数値計算した場合に, 計算モードは減衰する. なお, 物理モードは Euler 法と同様に von Neumann の条件を満足している.

6.4 宿題

振動型の微分方程式,

$$\frac{dx}{dt} = i\omega x, \quad (38)$$

に関して, Euler 法と Adams-Bashforth 法による計算の安定性を評価しなさい. ここで
行った解析から, (25) を解いた結果の図 4, 5 の違いについて考察しなさい.*⁶

参考文献

- 大関 誠, 「スペクトルモデル入門」, 気象研究ノート 第 211 号, 日本気象学会,
2006 年.
- 川上 一郎, 「数値計算」, 第 6 章, 岩波書店, 1989 年.

⁶ λ は一般に複素数なので, $|\lambda| = \sqrt{\lambda\lambda^}$ であることに気をつけなさい. (25) は (38) のような振動型の微分方程式である. なぜなら, (25) の解は $\omega = \sqrt{k/m}$ の振動数で振動するからである.